

## Creating a Distance Matrix from Census Data with SAS<sup>1</sup>

Many summary census files have information about the longitude and latitude of the center geographic units. These can be used to calculate the distance between the centers geographic units.

Following is a SAS program that uses data from ICPSR study 6054 (Census of Population and Housing, 1990 [United States]: Summary Tape File 3C) to calculate the distance between the centers of states. The data set used has 51 cases (50 states plus DC), and this N of Cases is reflected in the array statements that create variables (for example: `array dlon dlon1 - dlon51 ;`). If a different level of geography is used (for example, counties) then the number of elements in each array must be modified to reflect the number of cases.

```
** distance_matrix ;

* get longitude and latitude in decimal degrees - from study 6054;
title 'distance_matrix' ;
LIBNAME LIBRARY 'C:\all\help\helpnew\distance\SASLIB' ;
options linesize = 80 pagesize = 70 ;
data longlat ;
infile 'C:\all\help\helpnew\distance\state.data' ;
MERGVAR=1 ;
** varvar is created here just to show how a variable of interest ;
** would be used. This variable could be %urban, for example ;
varvar = ranuni(333) ;
label varvar = 'A variable of interest' ;
input stab $ statefip INTPTLAT INTPTLNG ; run ;

PROC TRANSPOSE DATA= longlat OUT=stabdat PREFIX= st ; VAR stab ; run ;
PROC TRANSPOSE DATA= longlat OUT=Y PREFIX= Y ; VAR INTPTLNG ; run ;
PROC TRANSPOSE DATA= longlat OUT=X PREFIX= X ; VAR INTPTLAT ; run ;
PROC TRANSPOSE DATA= longlat OUT=v PREFIX= b ; VAR varvar ; run ;

DATA xyv ; MERGE stabdat Y X V;
MERGVAR=1 ;

data LIBRARY.DISTANCE_MATRIX ; merge longlat xyv ; by MERGVAR ;

lngcur = (INTPTLNG * 3.1415926535 ) / 180 ;
latcur = (INTPTLAT * 3.1415926535 ) / 180 ;

** lat of other states ;
ARRAY X X1-X51;
** long of other states ;
ARRAY Y Y1-Y51;
** array for longitude (empty now) ;
ARRAY LNG LNG1-LNG51;
** array for latitude (empty now) ;
ARRAY LAT LAT1-LAT51;
** array for differences between longitudes (empty now) ;
array dlon dlon1 - dlon51 ;
** array for differences between latitudes (empty now) ;
array dlat dlat1 - dlat51 ;
** array for distance (empty now) ;
ARRAY DIS DIS1-DIS51;
```

---

<sup>1</sup>Prepared by Patty Glynn, University of Washington. 5/17/ 2004, revised 10/17/2006 C:\all\help\helpnew\distance\dist\_matrix.wpd

```
* http://www.census.gov/cgi-bin/geo/gisfaq?Q5.1  
* Haversine Formula (from R.W. Sinnott, "Virtues of the Haversine",  
* Sky and Telescope, vol. 68, no. 2, 1984, p. 159): ;
```

```
do over DIS ;  
LNG = (Y * 3.1415926535 ) / 180 ;  
LAT = (X * 3.1415926535 ) / 180 ;  
dlon = lngcur - lng ;  
dlat = latcur - lat ;  
a = (sin(dlat /2))**2  
    + cos(LAT)  
    * cos(latcur)  
    * (sin(dlon/2))**2 ;  
c = 2 * arsin(min(1,sqrt(a))) ;  
DIS = 3956*c ;  
END ;  
drop a c y1-y51 x1-x51 lng1-lng51 LAT1-LAT51 dlon1-dlon51 dlat1-dlat51  
mergvar _name_ _label_ INTPTLAT INTPTLNG lngcur latcur varvar B1-B51 ;  
  
proc print uniform ; id stab ; var st1 dis1 ; format dis1-dis51 6.0 ; RUN ;  
proc print uniform ; id stab ; var st2 dis2 ; format dis1-dis51 6.0 ; RUN ;  
proc print uniform ; id stab ; var st3 dis3 ; format dis1-dis51 6.0 ; RUN ;  
proc print uniform ; id stab ; var st4 dis4 ; format dis1-dis51 6.0 ; RUN ;  
proc print uniform ; id stab ; var st48 dis48 ; format dis1-dis51 6.0 ; RUN ;  
proc print uniform ; id stab ; var st51 dis51 ; format dis1-dis51 6.0 ; RUN ;
```

The program and data for testing can be downloaded from:

<http://staff.washington.edu/glynn/state.data>

[http://staff.washington.edu/glynn/distance\\_matrix.sas](http://staff.washington.edu/glynn/distance_matrix.sas)

On the next page you will find output from the above program (put into excel for easier viewing) so that distances can be checked.

